Implementing a solution to the generalised word problem for the Hilden group

Stephen Tawn

22 June 2009

Saul Schleimer[4] has shown that the membership problem for the mapping class group of a handlebody inside the mapping class group of its boundry is solvable in polynomial time. We will give a slight modification of this argument to show that the membership problem for the Hilden (or wicket) group inside the braid group is solvable and implement the algorithm in MAGMA[1]. This is a literate MAGMA document [6] and contains the complete MAGMA code.

Fix $n > 0$ and let B_{2n} be the braid group on $2n$ strings.

```
if not assigned n then n := 3; end if;
if not assigned B then B := BraidGroup(2*n); end if;
```

Load a fix for a bug with the ***hom*** constructor.

```
load "hom.m";
```

Let $\mathbb{B}^3_+$ be half a unit ball in $\mathbb{R}^3$, ie the intersection of the unit ball $\mathbb{B}^3$ with the halfspace $\mathbb{R}^3_+ = \{z \geq 0\}$. Let a be n unknotted arcs in $\mathbb{B}^3_+$ such that the boundary of each arc lies in $\mathbb{R}^2$. The Hilden group H_{2n} is the orientation preserving mapping class group of $\mathbb{B}^3_+$ fixing a setwise and $\partial\mathbb{B}^3_+ \setminus \mathring{\mathbb{B}}^2$ pointwise. The inclusion $i\colon (\mathbb{B}^2, \partial a, \partial\mathbb{B}^2) \to (\mathbb{B}^3_+, a, \partial\mathbb{B}^3_+ \setminus \mathring{\mathbb{B}}^2)$ induces the embedding $H_{2n} \hookrightarrow B_{2n}$. A generating set for a similar group was found by Hilden[3] and a presentation for H_{2n} was calculated independantly by Brendle-Hatcher[2] and the author[5].

Pick a point P on $\partial\mathbb{B}^2$, let $F = \pi_1(\mathbb{B}^2 \setminus \partial a, P)$ be the fundamental group of $\mathbb{B}^2 \setminus \partial a$ and let $G = \pi_1(\mathbb{B}^3_+ \setminus a, P)$ be the fundamental group of $\mathbb{B}^3_+ \setminus a$. The group F is isomorphic to the free group of rank $2n$ and G is isomorphic to the free group of rank n. (We will represent the elements of F by straight line programs.)

```
F := SLPGroup(2*n);
G := FreeGroup(n);
```

MSC2000. Primary: 20–04; Secondary: 20F10, 20F36, 20F38, 57M07

Key words and phrases. Hilden group, generalised word problem, MAGMA

The inclusion map i induces a map $\phi\colon F \to G$. If we pick paths $x_1, x_2, \ldots, x_n$ and $y_1, y_2, \ldots y_n$ in $\mathbb{B}^2$ as in Figure 1 then F is generated by $x_1, x_2, \ldots x_n$ and $y_1, y_2, \ldots y_n$, G is generated by $z_1, z_2, \ldots z_n$ where z_i is the image of x_i in G. The map ϕ is given by $\phi(x_i) = z_i$ and $\phi(y_i) = 1$.

```
x := [ F.i : i in [1 .. n] ];
y := [ F.(n + i) : i in [1 .. n] ];
z := [ G.i : i in [1 .. n] ];

φ := HOM( F → G, [ x[i] → z[i] : i in [1..n] ]
                 cat [ y[i] → ID(G) : i in [1..n] ] );
```



Figure 1: Generators of F and G

Viewing the braid group B_{2n} as the mapping class group of the puctured disc $\mathbb{B}^2 \setminus \partial a$ we have a right action of the braid group on the free group F. If we let σ_i be a clockwise half-twist interchanging the ith and $i+1$st points of ∂a then this action is given by the following.

$$x_i \cdot \sigma_{2j-1} = \begin{cases} y_i\, x_i^{-1} & \text{for } j = i \\ x_i & \text{for } j \neq i \end{cases} \qquad x_i \cdot \sigma_{2j} = \begin{cases} x_{i-1}^{-1}\, y_{i-1} & \text{for } i = j+1 \\ x_i & \text{for } i \neq j+1 \end{cases}$$

$$y_i \cdot \sigma_{2j-1} = y_i \qquad y_i \cdot \sigma_{2j} = \begin{cases} y_i\, x_{i+1}\, y_i^{-1}\, x_i & \text{for } j = i \\ x_{i-1}^{-1}\, y_{i-1}\, x_i^{-1}\, y_i & \text{for } j+1 = i \\ y_i & \end{cases}$$

```
odd := func< j | HOM( F → F,
                    [ x[j] → y[j] * x[j]^-1 ]
                    cat [ x[i] → x[i] : i in [1..n] | i ne j ]
                    cat [ y[i] → y[i] : i in [1..n] ]
               ) >;

even := func< j | HOM( F → F,
                    [ x[j+1] → x[j]^-1 * y[j] ]
                    cat [ x[i] → x[i] : i in [1..n] | i ne j + 1 ]
                    cat [ y[j] → y[j] * x[j+1] * y[j]^-1 * x[j] ]
                    cat [ y[j+1] → x[j]^-1*y[j]*x[j+1]^-1*y[j+1] ]
                    cat [ y[i] → y[i] : i in [1..n]
                                   | (j ne i) and (j+1 ne i)]
               ) >;
```

With the inverses as follows.

$$x_i \cdot \sigma_{2j-1}^{-1} = \begin{cases} x_i^{-1}\, y_i & \text{for } j = i \\ x_i & \text{for } j \neq i \end{cases} \qquad x_i \cdot \sigma_{2j}^{-1} = \begin{cases} x_i^{-1}\, x_{i-1}^{-1}\, y_{i-1}\, x_i & \text{for } i = j+1 \\ x_i & \text{for } i \neq j+1 \end{cases}$$

$$y_i \cdot \sigma_{2j-1}^{-1} = y_i \qquad y_i \cdot \sigma_{2j}^{-1} = \begin{cases} x_i\, x_{i+1} & \text{for } j = i \\ x_i^{-1}\, x_{i-1}^{-1}\, y_{i-1}\, y_i & \text{for } j+1 = i \\ y_i & \end{cases}$$

```
oddBar := func< j | HOM( F → F,
                    [ x[j] → x[j]^-1 * y[j] ]
                    cat [ x[i] → x[i] : i in [1..n] | i ne j ]
                    cat [ y[i] → y[i] : i in [1..n] ]
               ) >;

evenBar := func< j | HOM( F → F,
                    [ x[j+1] → x[j+1]^-1 * x[j]^-1 * y[j] * x[j+1] ]
                    cat [ x[i] → x[i] : i in [1..n] | i ne j + 1 ]
                    cat [ y[j] → x[j] * x[j+1] ]
                    cat [ y[j+1] → x[j+1]^-1*x[j]^-1*y[j]*y[j+1] ]
                    cat [ y[i] → y[i] : i in [1..n]
                                   | (j ne i) and (j+1 ne i)]
                  ) >;
```

We can put these automorphisms in to two sequences, $S = [\sigma_1, \ldots, \sigma_{2n-1}]$ and $\bar{S} = [\sigma_1^{-1}, \ldots, \sigma_{2n-1}^{-1}]$.

```
S := [ IsEven(i) select even(i div 2)
```

```
                    else odd(i div 2 + 1) : i in [1..2*n-1] ];
SBar := [ IsEven(i) select evenBar(i div 2)
                    else oddBar(i div 2 + 1) : i in [1..2*n-1] ];
```

We can represent the elements of the braid group as sequences of integers. The sequence $[k_1, k_2, \ldots, k_N]$ for $k_i \in \{\pm 1, \pm 2, \ldots \pm (2n-1)\}$ represents the braid $\sigma_{k_1}\sigma_{k_2}\cdots\sigma_{k_N}$ where for negative k we define $\sigma_k = \sigma_{-k}^{-1}$. The action can now be represented as follows.

```
action := func<x, i | (i ge 0) select S[i](x) else SBar[-i](x)>;
```

We have some basic tests to check that every thing is working correctly. We can check the inverses and the braid relations.

```
function testInverses()
    Fprime := FreeGroup(2*n);
    evaluate := hom< F -> Fprime | [ Fprime.i : i in [1..2*n] ] >;
    X := [ evaluate( S[j]( SBar[j]( F.i ) ) ) eq evaluate( F.i )
                 : i in [1..n], j in [1..n-1] ];
    Y := [ evaluate( SBar[j]( S[j]( F.i ) ) ) eq evaluate( F.i )
                 : i in [1..n], j in [1..n-1] ];
    return &and (X cat Y);
end function;

function testRelations()
    Fprime := FreeGroup(2*n);
    evaluate := hom< F -> Fprime | [ Fprime.i : i in [1..2*n] ] >;
    // Commutivity relations
    X := [ evaluate((A * B)(F.i)) eq evaluate((B * A)(F.i))
                  where A := S[j]
                  where B := S[k]
               : i in [1..n], j in [1..k-2], k in [1..n-1] ];
    // Braid realtaions
    Y := [ evaluate((A * B * A)(F.i)) eq evaluate((B * A * B)(F.i))
                  where A := S[j]
                  where B := S[j+1]
               : i in [1..n], j in [1..n-1] ];
    return &and (X cat Y);
end function;

procedure test()
    print "Testing inverses:\t", testInverses();
    print "Testing realtions:\t", testRelations();
end procedure;
```

Theorem 1. *A braid $b \in B_{2n}$ is in the Hilden group if and only if for each $i = 1, \ldots, n$ we have $\phi(y_i \cdot b) = 1$.*

```
function inHilden(braid)
    Y := y;
    for i in ElementToSequence(braid) do
        Y := action(Y, i);
    end for;
    return &and [ φ(y) eq Id(G) : y in Y ];
end function;
```

Proof. It is clear that every element of the Hilden group will take any loop in $\mathbb{B}^2 \setminus \partial a$ that is null-homotopic in $\mathbb{B}^3_+ \setminus a$ to a loop that is null-homotopic in $\mathbb{B}^3_+ \setminus a$.

Now suppose that $b \in B_{2n}$ is a braid and that for each $i = 1, \ldots, n$ we have $\phi(y_i \cdot b) = 1$. Pick a map $\beta\colon \mathbb{B}^2 \to \mathbb{B}^2$ representing b and loops Y_i representing y_i. By Dehn's lemma, we can pick discs D_i and D'_i in $\mathbb{B}^3_+ \setminus a$ such that $Y_i = \partial D_i$ and $\beta(Y_i) = \partial D'_i$. The map $\beta\colon Y_i \to B(Y_i)$ gives a homeomorphism of the boundary of a disc, and hence can be extended to a homeomorphims of the whole disc. So we now have a homeomorphism $\beta\colon \mathbb{B}^2 \cup \bigcup_i D_i \to \mathbb{B}^2 \cup \bigcup_i D'_i$.

The discs $\mathbb{B}^2 \cup \bigcup_i D_i$ separate B^3_+ into n balls $\mathcal{B}_1, \mathcal{B}_2, \ldots, \mathcal{B}_n$ each containing one arc and one solid ball $\mathcal{B}$. Similarly, $\mathbb{B}^2 \cup \bigcup_i D'_i$ gives balls $\mathcal{B}'_1, \mathcal{B}'_2, \ldots, \mathcal{B}'_n$ and $\mathcal{B}'$.

As β is the identity on $\partial\mathbb{B}^2$ we can extend β so that it is the identity on $\partial B^3_+ \setminus \mathring{\mathbb{B}}^2$. Now β gives a homeomorphism of $\partial\mathcal{B}$ to $\partial\mathcal{B}'$ and so can be extended to a homeomorphism $\mathcal{B} \to \mathcal{B}'$.

It remains to deal with the balls $\mathcal{B}_i$. The map β gives a homeomorphism $\partial\mathcal{B}_i \to \partial\mathcal{B}'_i$ and this can be extended to a homeomorphism $\mathcal{B}_i \to \mathcal{B}'_i$. The image of the arc a_i under this map will be ambiant isotopic rel $\partial\mathcal{B}'_i$ to the arc in $\mathcal{B}'_i$. So we may assume that we choose our extension so that it takes the arc to the arc.

Hence we have extended β to a map

$$B\colon (\mathbb{B}^3_+, a, \partial\mathbb{B}^3_+ \setminus \mathring{\mathbb{B}}^2) \to (\mathbb{B}^3_+, a, \partial\mathbb{B}^3_+ \setminus \mathring{\mathbb{B}}^2).$$

Therefore b is in the Hilden group. □

References

[1] W. Bosma, J. Cannon, and C. Playoust. The Magma Algebra System I: The User Language. *Journal of Symbolic Computation*, 24(3-4):235–265, 1997.

[2] T. Brendle and A. Hatcher. Configuration spaces of rings and wickets. arXiv:0805.4354.

[3] Hugh M. Hilden. Generators for two groups related to the braid group. *Pacific J. Math.*, 59(2):475–486, 1975.

[4] Saul Schleimer. Polynomial-time word problems. *Commen. Math. Helv.*, 83:741–765, 2008.

[5] Stephen Tawn. A presentation for hilden's subgroup of the braid group. arXiv:0706.4421. to appear in Math. Res. Lett.

[6] Don Tayloy. Literate Magma programming. `http://www.maths.usyd.edu.au/u/don/code/Magma/magmatex.html`.